# Sketch a cone and its base

A cone is the union of a set of half-lines that start at a common apex point and go through a base which can be any parametric curve. Show that the graph of z = ( x ^2 + 4y ^2)^(1/2) is a cone. What can be chosen as its base? Sketch the base.

I can see how I show this is cone by hand, just squaring and dividing across by 4 to get (z^2)/4=(x^2)/4 + y^2 but not sure how I can sketch it mathematica, including the base. Do I have to change it from this form? Is it Plot 3-D or is it parametric plot as I'll need to use something as a base. Anyone got any advice?

• Plot3D[(z^2)/4= (x^2)/4 + y^2), {x, -5, 5}, {y, -5, 5}, {z, -5, 5}] is returning error for the brackets? anyone know why this code wont plot this cone? – user24907 Feb 5 '15 at 19:51
• you mean something like ContourPlot3D[ x^2/4 + y^2 == z^2/4, {x, -5, 5}, {y, -5, 5}, {z, -5, 0}]? – kglr Feb 5 '15 at 20:00
• Perfect! just still unsure about when I should use ContourPlot3D or Plot3D or Parametric plot. So if I want to now add a parametric curve as base what function must I use? – user24907 Feb 5 '15 at 20:05
• mathopenref.com/coordparamellipse.html – Manuel --Moe-- G Feb 5 '15 at 20:08
• See e.g. Cone image refinement – Artes Feb 5 '15 at 20:29

cone1 = (x^2 + 4 y^2)^(1/2)

Plot3D[cone1, {x, -3, 3}, {y, -3, 3}]

ContourPlot[cone1, {x, -3, 3}, {y, -3, 3}, Contours -> {1, 2, 3}]

base1 = #^2 & /@ (cone1 == 1)

ContourPlot[Evaluate[base1], {x, -1, 1}, {y, -1, 1}]

parametricBase1 = {x -> Cos[t], y -> (1/2) Sin[t]}

base1 /. parametricBase1

coneXPlane = cone1 /. {x -> 0}

Plot[coneXPlane, {y, -3, 3}]

coneYPlane = cone1 /. {y -> 0}

Plot[coneYPlane, {x, -3, 3}]


• Thank you, do you mind explaining what this base function is what the symbols you've listed do? And why you haven't used z^2 in your cone expression? – user24907 Feb 5 '15 at 20:12
• less about doing your homework for you, more about examples of "natural" ways to refer to things in MMA, also added parametric form of base – Manuel --Moe-- G Feb 5 '15 at 20:18
• Yes I'm trying to understand of these "natural" ways. mainly confused as to #^2 &/ @ in the base function is. Is there any particular reason for theses variables? – user24907 Feb 5 '15 at 20:24
options = {PlotRange -> {{-5, 5}, {-5, 5}, {-5, 0}}, Mesh -> None,
ColorFunction -> "Rainbow", BaseStyle -> Opacity[.8], BoxRatios -> 1, Lighting -> "Neutral"};

ParametricPlot3D[{Cos[t] z, Sin[t] z/2, z}, {t, 0, 2 \[Pi]}, {z, -5,  0}, Evaluate@options]


ParametricPlot3D[z {Cos[t], Sin[2 t]/2, 1}, {t, 0, 2 \[Pi]}, {z, -5, 0}, Evaluate@options]


ParametricPlot3D[ z {Cos[t], Cos[3 t] Sin[t]/2, 1}, {t, 0, 2 \[Pi]}, {z, -5, 0},
Evaluate@options]


ParametricPlot3D[z { Cos[t] Cos[2 t], Sin[ t] Cos[ 2 t], 1}, {t, 0, 2 \[Pi]}, {z, -5, 0},
Evaluate@options]


• Thank you, so i would put this after ContourPlot3D[ x^2/4 + y^2 == z^2/4, {x, -5, 5}, {y, -5, 5}, {z, -5, 0}] ? all in a pair of [] brackets? – user24907 Feb 5 '15 at 20:26
• @user24907, the posted examples show how to use ParametricPlot3D to get cones with various parametric bases. – kglr Feb 5 '15 at 20:56

A simpler/improved parametrization (t,h) for an elliptic cone. It is more convenient from apex as it is at the apex that all generators converge.

 ParametricPlot3D[h {2 Sin[t],Cos[t],2 },{t,0,2 Pi},{h,0,5},PlotStyle-> {Yellow}]